3. Spectral averages.

To make comparisons with the observed spectra, the results of this high resolution calculation are averaged using a triangular window function to the spectral resolution of the spectrometer during measurements, e.g. 1 cm^{-1}. The triangular window function $WN(\nu;\nu_c,\Delta\nu)$ is given by

$$WN(\upsilon;\upsilon_c,\Delta\upsilon) = 1 - \frac{|\upsilon - \upsilon_c|}{\Delta\upsilon}, \quad (\upsilon_c - \Delta\upsilon) \le \upsilon \le (\upsilon_c + \Delta\upsilon) \quad , \tag{9}$$

where ν_c is the wavenumber where the emission is desired, and $\Delta\nu$ is the desired spectral resolution. The spectrally averaged emission spectra $I_{spec}(\nu_c)$ is obtained from

$$I_{spec}(\upsilon_c) = \frac{\sum_k WN(\upsilon_k;\upsilon_c,\Delta\upsilon) I_{spec}(\upsilon_k)}{\sum_k WN(\upsilon_k;\upsilon_c,\Delta\upsilon)} \quad , \tag{10}$$

and can be compared with measurements.

A section of code that implements spectral smoothing is below:

```
* Spectral smoothing subroutine.
* INPUTS:
* del=desired spectral resolution.
* wn = array containing wavenumbers used in high resolution calculation.
* wnave = array of wavenumbers in the averaged spectrum.
* spec = array containing radiance from the high resolution calculation.
* OUTPUT:
* specave = array of radiance values in the averaged spectrum.
* NOTES:
* del, wn, and wnave should be in the same units, e.g. 1/cm.
* spec and specave will be in the same units.
      SUBROUTINE SpectralSmooth(del,wn,spec,  wnave, specave)
      IMPLICIT REAL (a-h,o-z)
      REAL wn(452940),spec(452940),wnave(779), specave(779)
* Set your array dimensions to what you use in your problem.
      EXTERNAL win

      DO 30 j=1,779
      Ts=0.
      ws=0.

      DO 40 i=1,452940
      IF (wn(i).GE.wnave(j)-del) THEN
        IF (wn(i).LE.wnave(j)+del) THEN
         wnt=wn(i)
         wnte=wnave(j)
         delt=del
         window=win(wnt,wnte,delt)
         Ts=Ts + window*spec(i)  ! Temporary spectrum value.
         ws=ws + window
        ELSE
         GOTO 35 ! Have got all the bin filled with spectra.
        END IF
      END IF
40    CONTINUE

35    Ts = Ts / ws
30    specave(j)=Ts
      END

* Trianglular window function to use for spectral smoothing.
* wn is the desired wavenumber for the window.
* wnc is the center wavenumber of the bin.
* del is the FWHM of the spectral bin.
* All inputs should be in the same units.
      REAL FUNCTION win(wn,wnc,del)
      IMPLICIT REAL (a-h,o-z)

      IF (wn.EQ.wnc) THEN
       win=1.
      ELSE IF ((wn.GE.wnc-del).AND.(wn.LT.wnc)) THEN
       win=1.-(wnc-wn)/del
      ELSE IF ((wn.GT.wnc).AND.(wn.LE.wnc+del)) THEN
       win=1.-(wn-wnc)/del
      ELSE
       win=0.
      END IF
      RETURN
      END
```